Computer Sciences Department
University of Wisconsin
1210 West Dayton Street
Madison, Wisconsin 53706

DISCRETE NEWTONIAN GRAVITATION AND THE THREE-BODY PROBLEM

by

Donald Greenspan

Appendix: Fortran Program for the Three-Body Problem

by

Alvin B. Schubert

Technical Report #133
October 1971

Discrete Newtonian Gravitation and the Three-Body Problem

by

Donald Greenspan

ABSTRACT

Newtonian gravitation is studied from a discrete point of view in that the dynamical equation is an energy conserving difference equation. Application is made to planetary type, nondegenerate three-body problems and several computer examples of perturbed orbits are given.

1. Introduction

The dynamical behavior of n interacting bodies has long been of major interest in science and mathematics (see, e.g., refs. [1]-[10], [12]-[15], and the additional references contained therein). Typical important n-body problems occur in the study of the solar system under the usual assumptions that n be relatively small and that capture, but not collision, be admissible, and in the study of Brownian motion under the usual assumptions that n be relatively large and that collisions occur in accordance with an assumed probabilistic distribution.

In this paper we will study an orbit type problem which is of basic importance in astronomy. More precisely, we will study the dynamical behavior of three nondegenerate bodies acted upon mutually by the force of gravitation. Our model will be computer oriented in the sense that the dynamical equations will be energy conserving difference equations which can be solved directly by high-speed arithmetic. For clarity, the discussion will be given in two dimensions, though the method extends easily, and in a natural way, to n bodies in any number of dimensions.

2. Discrete Newtonian Gravitation

For $\Delta t > 0$ and $t_k = k\Delta t$, $k = 0,1,2,\ldots$, and for each of $i = 1,2,3$, let particle P_i of mass m_i be located at $(x_{i,k}, y_{i,k})$, have velocity $\vec{v}_{i,k} = (v_{i,k,x}, v_{i,k,y})$, and acceleration $\vec{a}_{i,k} = (a_{i,k,x}, a_{i,k,y})$ at time t_k. Let position, velocity and acceleration be related by

$$(2.1)\qquad \frac{v_{i,k+1,x} + v_{i,k,x}}{2} = \frac{x_{i,k+1} - x_{i,k}}{\Delta t}, \quad i=1,2,3;\ k=0,1,2,\ldots$$

$$(2.2)\qquad \frac{v_{i,k+1,y} + v_{i,k,y}}{2} = \frac{y_{i,k+1} - y_{i,k}}{\Delta t}, \quad i=1,2,3;\ k=0,1,2,\ldots$$

$$(2.3)\qquad a_{i,k,x} = \frac{v_{i,k+1,x} - v_{i,k,x}}{\Delta t}, \quad i=1,2,3;\ k=0,1,2,\ldots$$

$$(2.4)\qquad a_{i,k,y} = \frac{v_{i,k+1,y} - v_{i,k,y}}{\Delta t}, \quad i=1,2,3;\ k=0,1,2,\ldots .$$

To relate force and acceleration, let us assume a discrete Newtonian equation of the form

$$(2.5)\qquad \vec{F}_{i,k} = m_i \vec{a}_{i,k}; \quad i=1,2,3;\ k=0,1,2,\ldots ,$$

with

$$(2.6)\qquad \vec{F}_{i,k} = (F_{i,k,x}, F_{i,k,y})$$

and

(2.7)
$$F_{1,k,x} = -\frac{Gm_1m_2[(x_{1,k+1}+x_{1,k})-(x_{2,k+1}+x_{2,k})]}{r_{12,k}r_{12,k+1}(r_{12,k}+r_{12,k+1})} - \frac{Gm_1m_3[(x_{1,k+1}+x_{1,k})-(x_{3,k+1}+x_{3,k})]}{r_{13,k}r_{13,k+1}(r_{13,k}+r_{13,k+1})}$$

(2.8)
$$F_{1,k,y} = -\frac{Gm_1m_2[(y_{1,k+1}+y_{1,k})-(y_{2,k+1}+y_{2,k})]}{r_{12,k}r_{12,k+1}(r_{12,k}+r_{12,k+1})} - \frac{Gm_1m_3[(y_{1,k+1}+y_{1,k})-(y_{3,k+1}+y_{3,k})]}{r_{13,k}r_{13,k+1}(r_{13,k}+r_{13,k+1})}$$

(2.9)
$$F_{2,k,x} = -\frac{Gm_1m_2[(x_{2,k+1}+x_{2,k})-(x_{1,k+1}+x_{1,k})]}{r_{12,k}r_{12,k+1}(r_{12,k}+r_{12,k+1})} - \frac{Gm_2m_3[(x_{2,k+1}+x_{2,k})-(x_{3,k+1}+x_{3,k})]}{r_{23,k}r_{23,k+1}(r_{23,k}+r_{23,k+1})}$$

(2.10)
$$F_{2,k,y} = -\frac{Gm_1m_2[(y_{2,k+1}+y_{2,k})-(y_{1,k+1}+y_{1,k})]}{r_{12,k}r_{12,k+1}(r_{12,k}+r_{12,k+1})} - \frac{Gm_2m_3[(y_{2,k+1}+y_{2,k})-(y_{3,k+1}+y_{3,k})]}{r_{23,k}r_{23,k+1}(r_{23,k}+r_{23,k+1})}$$

$$(2.11) \qquad F_{3,k,x} = -\frac{Gm_1m_3[(x_{3,k+1}+x_{3,k})-(x_{1,k+1}+x_{1,k})]}{r_{13,k}r_{13,k+1}(r_{13,k}+r_{13,k+1})}$$

$$-\frac{Gm_2m_3[(x_{3,k+1}+x_{3,k})-(x_{2,k+1}+x_{2,k})]}{r_{23,k+1}r_{23,k}(r_{23,k}+r_{23,k+1})}$$

$$(2.12) \qquad F_{3,k,y} = -\frac{Gm_1m_3[(y_{3,k+1}+y_{3,k})-(y_{1,k+1}+y_{1,k})]}{r_{13,k}r_{13,k+1}(r_{13,k}+r_{13,k+1})}$$

$$-\frac{Gm_2m_3[(y_{3,k+1}+y_{3,k})-(y_{2,k+1}+y_{2,k})]}{r_{23,k+1}r_{23,k}(r_{23,k}+r_{23,k+1})},$$

where $r_{ij,k}$ is the distance between P_i and P_j at time t_k. Gravitation law (2.6)-(2.12) is a discrete "$\frac{1}{r^2}$ law" of attraction.

From any given set of initial data $(x_{i,0}, y_{i,0})$ and $\vec{v}_{i,0}$, $i = 1,2,3$, collisionless motions of P_1, P_2 and P_3 are determined by (2.1)-(2.12). However, before discussing the details of how to use a digital computer to generate these motions, let us show that our discrete formulation is energy conserving, since one can expect physical stability from a three-body system which has no supply of new energy. It is of fundamental importance to note that the usual models generated by direct differencing of the continuous three-body equations are <u>not</u> energy conserving.

The work W_i done by $\vec{F}_{i,k}$ on P_i from time t_0 to time t_n is defined by

$$(2.13) \qquad W_i = \sum_{k=0}^{n-1} [(x_{i,k+1}-x_{i,k})F_{i,k,x} + (y_{i,k+1}-y_{i,k})F_{i,k,y}],$$

while the total work W done on the system is defined by

$$(2.14) \qquad W = \sum_{i=1}^{3} W_i \; .$$

From (2.1), (2.3), (2.5) and (2.6), one has first that

$$\sum_{k=0}^{n-1} [(x_{i,k+1}-x_{i,k})F_{i,k,x}] = \sum_{k=0}^{n-1} [(x_{i,k+1}-x_{i,k})m_i a_{i,k,x}]$$

$$= m_i \sum_{k=0}^{n-1} [\frac{(x_{i,k+1}-x_{i,k})}{\Delta t} (v_{i,k+1,x}-v_{i,k,x})]$$

$$= \frac{m_i}{2} \sum_{k=0}^{n-1} [(v_{i,k+1,x}+v_{i,k,x})(v_{i,k+1,x}-v_{i,k,x})]$$

$$= \frac{m_i}{2} \sum_{k=0}^{n-1} [v_{i,k+1,x}^2 - v_{i,k,x}^2]$$

$$= \frac{m_i}{2} (v_{i,n,x}^2 - v_{i,0,x}^2) \; .$$

Thus,

$$(2.15) \qquad \sum_{k=0}^{n-1} [(x_{i,k+1}-x_{i,k})F_{i,k,x}] = \frac{m_i}{2} v_{i,n,x}^2 - \frac{m_i}{2} v_{i,0,x}^2 \; .$$

Similarly,

$$(2.16) \qquad \sum_{k=0}^{n-1} [(y_{i,k+1} - y_{i,k}) F_{i,k,y}] = \frac{m_i}{2} v_{i,n,y}^2 - \frac{m_i}{2} v_{i,0,y}^2 .$$

Addition of (2.15) and (2.16) then yields

$$(2.17) \qquad \sum_{k=0}^{n-1} [(x_{i,k+1} - x_{i,k}) F_{i,k,x} + (y_{i,k+1} - y_{i,k}) F_{i,k,y}]$$

$$= \frac{m_i}{2} (v_{i,n,x}^2 + v_{i,n,y}^2) - \frac{m_i}{2} (v_{i,0,x}^2 + v_{i,0,y}^2).$$

Now, let the kinetic energy $K_{i,k}$ of particle P_i at t_k be defined by

$$(2.18) \qquad K_{i,k} = \frac{1}{2} m_i (v_{i,k,x}^2 + v_{i,k,y}^2) ,$$

and let the kinetic energy K_k of the system at time t_k be defined by

$$(2.19) \qquad K_k = \sum_{i=1}^{3} K_{i,k}.$$

Then (2.13), (2.14) and (2.17)-(2.19) imply

$$(2.20) \qquad W = K_n - K_0 .$$

Note that, in establishing (2.20), no special structure for $F_{i,k,x}$ and $F_{i,k,y}$ was ever used. Suppose then one substitutes (2.7)-(2.12) into (2.13). Then, simple, but tedious, calculation yields

$$W = -Gm_1m_2 \sum_{k=0}^{n-1} \left(\frac{r_{12,k+1} - r_{12,k}}{r_{12,k}\, r_{12,k+1}} \right)$$

$$-Gm_1m_3 \sum_{k=0}^{n-1} \left(\frac{r_{13,k+1} - r_{13,k}}{r_{13,k}\, r_{13,k+1}} \right)$$

$$-Gm_2m_3 \sum_{k=0}^{n-1} \left(\frac{r_{23,k+1} - r_{23,k}}{r_{23,k}\, r_{23,k+1}} \right)$$

$$= -Gm_1m_2 \left(\frac{1}{r_{12,0}} - \frac{1}{r_{12,n}} \right) - Gm_1m_3 \left(\frac{1}{r_{13,0}} - \frac{1}{r_{13,n}} \right)$$

$$-Gm_2m_3 \left(\frac{1}{r_{23,0}} - \frac{1}{r_{23,n}} \right) .$$

Defining the potential energy $V_{ij,k}$ of the pair P_i and P_j at t_k by

$$V_{ij,k} = -\frac{Gm_im_j}{r_{ij,k}}$$

implies, then, that

(2.21) $$W = V_{12,0} + V_{13,0} + V_{23,0} - V_{12,k} - V_{13,k} - V_{23,k},$$

while defining the potential energy V_k of the system at t_k by

(2.22) $$V_k = V_{12,k} + V_{13,k} + V_{23,k}$$

yields immediately, from (2.21),

(2.23) $$W = V_0 - V_n.$$

Elimination of W between (2.20) and (2.23) implies, finally,

$$K_n + V_n = K_0 + V_0. \tag{2.24}$$

Since n in (2.24) is arbitrary, it follows that the sum of the kinetic and potential energies is invariant with respect to time, which is, of course, the law of conservation of energy.

3. Solution of the Three-Body Problem

The computer implementation of an initial value problem for (2.1)-(2.12) can be described precisely as follows. The system (2.1)-(2.12) is rewritten in the more convenient, equivalent form

$$(3.1) \qquad x_{i,k+1} = x_{i,k} + \frac{\Delta t}{2}(v_{i,k+1,x} + v_{i,k,x}), \qquad i=1,2,3$$

$$(3.2) \qquad y_{i,k+1} = y_{i,k} + \frac{\Delta t}{2}(v_{i,k+1,y} + v_{i,k,y}), \qquad i=1,2,3$$

$$(3.3) \qquad v_{i,k+1,x} = v_{i,k,x} + \frac{\Delta t}{m_i} F_{i,k,x}, \qquad i=1,2,3$$

$$(3.4) \qquad v_{i,k+1,y} = v_{i,k,y} + \frac{\Delta t}{m_i} F_{i,k,y}, \qquad i=1,2,3,$$

where $F_{i,k,x}$ and $F_{i,k,y}$ are given by (2.7)-(2.12). Now, beginning with the initial data $x_{i,0}$, $y_{i,0}$, $v_{i,0,x}$, $v_{i,0,y}$, apply Newton's method to (3.1)-(3.4) to generate $x_{i,1}$, $y_{i,1}$, $v_{i,1,x}$ and $v_{i,1,y}$. Using these new results for initial data, apply Newton's method again to (3.1)-(3.4) to generate $x_{i,2}$, $y_{i,2}$, $v_{i,2,x}$ and $v_{i,2,y}$. Proceed in the indicated recursive fashion. Thus, for each value of $k = 0,1,2,\ldots$, the twelve equations (3.1)-(3.4) are solved for $x_{i,k+1}$, $y_{i,k+1}$, $v_{i,k+1,x}$ and $v_{i,k+1,y}$ by Newton's method using the results for t_k as initial data. Further, the initial guess for the Newtonian iteration is taken to be $x^{(0)}_{i,k+1} = x_{i,k}$, $y^{(0)}_{i,k+1} = y_{i,k}$, $v^{(0)}_{i,k+1,x} = v_{i,k,x}$, and $v^{(0)}_{i,k+1,y} = v_{i,k,y}$.

A complete Fortran program for the method described above is given in [11].

4. Examples

From the variety of examples run on the UNIVAC 1108 at the University of Wisconsin, we will now discuss several which illustrate the change of particle behavior as the masses and initial values are varied. In each example, the time step is $\Delta t = 0.001$ and cgs units are used, so that $G = (6.67)10^{-8}$.

Example 1.

Let P_1 be a body of mass $m_1 = (6.67)^{-1} 10^8$ which is located at $(0,0)$ and has zero initial velocity. Let P_2 be a body which is located at $(0.5, 0)$, which has velocity components $v_{2,0,x} = 0$, $v_{2,0,y} = 1.63$, and whose mass is negligible compared to that of P_1. Then, it follows from the usual methods of celestial mechanics that the trajectory of P_2 is an elliptic orbit with P_1 at a focus, with semi-major axis $a = 0.746$, and with period $\tau = 4.04$. Now, if the method of Section 3 is to be of any value, it should be applicable to the above problem and should yield high accuracy. To apply the method of Section 3, one simply sets $m_3 = 0$ in (2.7)-(2.10) and applies recursion formulas (3.1)-(3.4) with only $i = 1,2$. In this fashion, the motion of P_2 is given, specifically, by

$$(4.1) \qquad x_{2,k+1} = x_{2,k} + \frac{\Delta t}{2} (v_{2,k+1,x} + v_{2,k,x})$$

(4.2) $$y_{2,k+1} = y_{2,k} + \frac{\Delta t}{2} (v_{2,k+1,y} + v_{2,k,y})$$

(4.3) $$v_{2,k+1,x} = v_{2,k,x} - \Delta t(x_{2,k+1} + x_{2,k})(x_{2,k}^2 + y_{2,k}^2)^{-1/2} (x_{2,k+1}^2 + y_{2,k+1}^2)^{-1/2} [(x_{2,k}^2 + y_{2,k}^2)^{1/2} + (x_{2,k+1}^2 + y_{2,k+1}^2)^{1/2}]^{-1}$$

(4.4) $$v_{2,k+1,y} = v_{2,k,y} - \Delta t(y_{2,k+1} + y_{2,k})(x_{2,k}^2 + y_{2,k}^2)^{-1/2} (x_{2,k+1}^2 + y_{2,k+1}^2)^{-1/2} [(x_{2,k}^2 + y_{2,k}^2)^{1/2} + (x_{2,k+1}^2 + y_{2,k+1}^2)^{1/2}]^{-1}.$$

From the given initial data, the motion generated from (4.1)-(4.4) up to $t_{350,000} = 350$ consisted of 86+ orbits, the 86th of which is shown in Figure 4.1. For this particular orbit, the period was $\tau = 4.046$ and half the distance between the X intercepts was $a = 0.746$, which is in complete agreement with the analytical results described above. The total computing time was under five minutes.

Example 2.

The data of Example 1 were changed only by the selection of a new mass $m_2 = (6.67)^{-1} 10^6$ for P_2. This time, of course, the mass center of the system is in uniform motion and P_2 is still in orbit

FIG. 4.1



relative to P_1. This orbit, denoted by C_2 in Figure 4.2 has period $\tau = 3.901$ and semi-major axis $a = 0.730$. The orbit labeled C_1 in Figure 4.2 is that described in Example 1, so that the superposition of C_2 on C_1 shows that the increase in mass of P_2 has resulted in a smaller orbit. Because the present example consists of only two bodies, the above assertion can be verified directly from Newton's form of Kepler's third law:

$$(m_1 + m_2)\tau^2 = \frac{4\pi^2}{G} a^3 , \tag{4.5}$$

since

$$(m_1 + m_2)\tau^2 = [(6.67)^{-1} 10^8 + (6.67)^{-1} 10^6](3.901)^2 \sim (2.30)\,10^8 \tag{4.6}$$

and

$$\frac{4\pi^2}{G} a^3 = \frac{4\pi^2}{6.67 \cdot 10^{-8}} (0.730)^3 \sim (2.30)\,10^8 . \tag{4.7}$$

Additional analysis of this example will not be given because the next two examples, which are full three-body problems, have all the subtleties of the present one, and several in addition.

FIG. 4.2



Example 3.

Example 2 was modified by introducing the third particle P_3 of mass $m_3 = (6.67)^{-1} 10^5$, with initial position (-1,8), and with velocity components $v_{3,0,x} = 0$, $v_{3,0,y} = -3.75$. The initial arrangement of P_1, P_2 and P_3 is shown in Figure 4.3. The initial data were chosen so that P_2 and P_3 would arrive in the vicinity of (-1,0) almost simultaneously. In Figure 4.4 is shown the motion of P_1 from t_0 to t_{10000}. The positions marked with the integers $n = 0,1,2,\ldots,10$ are those of the particle at t_{1000n}. The motion indicates clearly the uniform motion of the mass center of the system, since the mass center is relatively close to the center of P_1. In Figure 4.5 is shown the motion of P_2 from t_0 to t_{5000} with the integers $n = 0,1,2,\ldots,5$ marking the positions t_{1000n}. In Figure 4.6 is shown the motion of P_3 from t_0 to t_{5000} with the same integer markings $n = 0,1,2,\ldots,5$ as for P_2. Particles P_2 and P_3 are closest at t_{2125} when P_2 is at (-0.9296, -0.1108) and P_3 is at (-0.9325, -0.1012). The effect of the gravitational interaction between P_2 and P_3 during their period of close proximity is to deflect P_2 outward, as is seen in Figure 4.5, and to deflect P_3 inward, as is seen in Figure 4.6. Moreover, after its first revolution about P_1, P_2 goes into the new orbit about P_1 which is shown in Figure 4.7. The end points of the major axis are (0.4943, 0.1664) and (-0.9105, -0.3075), so that $a = 0.74135$; the period is $\tau = 3.9905$; and the



P3
Gm3 = 0.001
v3,0 = (0,-3.75)
(0,8)
(0,6)
(0,4)
(0,2)
(-1,0)
(1,0)
P1
Gm1=1
v1,0 = (0,0)
P2
Gm2 = 0.01
v2,0 = (0,1.63)

FIG. 4.3



FIG. 4.4



FIG 4.5



FIG 4.6



FIG. 4.7

angle of inclination θ of the major axis with the X axis is given by $\tan\theta = 0.34$. Kepler's third law is again valid since

$$(m_1 + m_2)\,\tau^2 = \frac{10^8}{6.67}(1.01)\,(3.9905)^2 \sim (2.41)\,10^8,$$

while

$$\frac{4\pi^2}{G}\,(a)^3 = \frac{4\pi^2}{6.67\cdot 10^{-8}}\,(0.74135)^3 \sim (2.41)10^8.$$

Example 4.

Example 3 was modified only by increasing the mass of P_3 to $m_3 = (6.67) \cdot 10^6$, so that $m_2 = m_3$. The trajectories of P_1 and P_3 are similar to those described in Example 3. But this time the gravitational interaction between P_2 and P_3 is strong enough to pull P_2 out of its orbit. The trajectory of P_2 is shown from t_0 to t_{5000} in Figure 4.8.



FIG. 4.8

5. Remarks.

Example 3 of Section 4 demonstrates clearly, on a large scale, how an orbit can undergo rotation due only to gravitational forces. A similar example in which P_2 and P_3 come closest in the second quadrant yielded a rotation of P_2's orbit in which θ was negative. The basic implication is that perihelion motion can be positive or negative. Preliminary calculations of a Sun-Mercury-Venus model do show that the perihelion motion of Mercury, though small, is, at times positive, and at other times, negative. Details and further applications to problems in astronomy will be provided in a forthcoming paper.

References

1. B. J. Alder, "Studies in molecular dynamics III: A mixture of hard spheres", J. Chem. Phys., 40, 1964, p. 2724.

2. G. Birkhoff and R. E. Langer (Eds.), Orbit Theory, Proceedings of a Symposium in Applied Math., IX, Amer. Math. Soc., Providence, R.I., 1959.

3. W. J. Eckert, D. Brouwer and G. M. Clemence, "Coordinates of the five outer planets, 1653-2060," Astr. Papers Amer. Ephem., 12, 1951.

4. J. Gillespie and J. Nuttall (Eds.), Three-Particle Scattering in Quantum Mechanics, Benjamin, N. Y., 1968.

5. D. Greenspan, "Numerical studies of the 3-body problem," SIAM Jour. Appl. Math., 20, 1971, p. 67.

6. J. O. Hirschfelder, C. F. Curtiss and R. B. Bird, Molecular Theory of Gases and Liquids, Wiley, N. Y., 1954.

7. E. Leimanis, "Some recent advances in the dynamics of rigid bodies and celestial mechanics," Survey in Applied Math. II, Wiley, N. Y., 1958.

8. W. R. Longley, "A class of periodic orbits of an infinitesimal body subject to the attraction of n finite bodies," Trans. Amer. Math. Soc., 8, 1907, p. 159.

9. R. H. Miller and N. Alton, "Three dimensional n-body calculations," ICR Quart. Rpt. #18, Univ. Chicago, 1968.

10. L. M. Rauch and W. C. Riddell, "Iteration solutions of the analytical n-body problem," SIAM Jour. Appl. Math., 8, 1960, p. 568.

11. A. B. Schubert, "Fortran program for the three-body problem," Appendix, Tech. Rpt. #133, Dept. Comp. Sci., Univ. Wis., 1971.

12. E. Stromgren, "Connaissance actuelle des orbites dans le probleme des trois corps," Bull. Astr. (2), 9, 1935, p. 87.

13. J. E. Welch, F. H. Harlow, J. P. Shannon and B. J. Daly, "The MAC method," Tech. Rpt. #3425, Los Alamos Sci. Lab., Los Alamos, N.M., 1966.

14. A. Wintner, The Analytical Foundations of Celestial Mechanics, Princeton Univ. Press, Princeton, N.J., 1947.

15. J. Zumkley, "Ein numerisch gerechneter Specialfall des allgemeinen Dreikorperproblems in vereinfachter Behandlung," Astr. Nachr., 272, 1941, p. 66.

APPENDIX - FORTRAN PROGRAM FOR THE THREE-BODY PROBLEM
by A. B. Schubert

```
C       GENERAL DISCRETE THREE BODY PROBLEM WITH CONSERVATION
C       OF TOTAL ENERGY OF THE SYSTEM
C       GRAVITATIONAL FORCES ONLY CONSIDERED

C       ALL COMPUTATION DONE IN DOUBLE PRECISION


        IMPLICIT DOUBLE PRECISION (A-H,O-Z)
        DOUBLE PRECISION MASS(3)
        DIMENSION X0(3),Y0(3),VX0(3),VY0(3),X(3,3),Y(3,3),VX(3,3),VY(3,3),
       * FX(3),FY(3),R(3,2),GM(3)
        DATA MAXIT/150/,EPS/1.D-10/,G/1.D-2/

C       THE ABOVE DATA IS DEFINED AS FOLLOWS
C          MAXIT = MAX. NO. OF NEWTON ITERATIONS TO BE ALLOWED IN COM-
C                  PUTING POSITIONS AND VELOCITIES AT EACH TIME STEP
C            EPS = CONVERGENCE TOLERANCE IN NEWTON ITERATION
C              G = GRAVITATIONAL CONSTANT

     99 FORMAT(16D5.0)
     98 FORMAT(2I5,14D5.0)
     97 FORMAT('1DT ='D11.4,5X 'OMEGA ='D11.4,3X 'VX,VY =' 6D10.3/
       * 1X 'MASS =' 3D11.4/)
     96 FORMAT('0NEWTON ITERATION FAILED AFTER' I5,2X 'ITERATIONS')
     95 FORMAT('0NEWTON ITERATION FOR TIME STEP' I5)
     94 FORMAT(' R,FX,FY' 2X 9D13.6)
     93 FORMAT(1X 6D15.7)

C       READ PROBLEM-DEFINING PHYSICAL DATA
C          X0(I), Y0(I),I=1,2,3 = INITIAL PARTICLE POSITIONS
C         VX0(I),VY0(I),I=1,2,3 = INITIAL PARTICLE VELOCITIES
C               MASS(I),I=1,2,3 = PARTICLE MASSES

      3 READ(5,99,END=40) (X0(I),Y0(I),I=1,3),(VX0(I),VY0(I),I=1,3)
       * ,(MASS(I),I=1,3)

C       READ ADDITIONAL COMPUTATIONAL DATA
C           NMAX = MAXIMUM NO. OF TIME STEPS  FOR THIS DATA CASE
C          INCPR = TIME STEP INCREMENT FOR PRINTING OF RESULTS
C          OMEGA = OVERRELAXATION FACTOR IN NEWTON'S METHOD
C             DT = TIME STEP SIZE
C            END = INPUT CONTROL VARIABLE.
C                  END.EQ.0 IMPLIES MORE CARDS OF THIS TYPE WILL FOLLOW
C                  FOR THE SAME PHYSICAL DATA INPUT ABOVE.
C                  END.NE.0 IMPLIES NOT.

      5 READ(5,98)          NMAX,INCPR,OMEGA,DT,END
        WRITE(6,97) DT,OMEGA,(VX0(I),VY0(I),I=1,3),(MASS(I),I=1,3)

C       COMPUTE STATIC DATA-DEPENDENT PROGRAM VARIABLES AND INITIALIZE
C       DYNAMIC POSITION AND VELOCITY VECTORS

C       DEFINITIONS OF X,Y,VX,VY ARRAYS
C
```

```
C       FOR I=1,2,3
C          X(I,1) = X-COMPONENT OF POSITION OF PARTICLE I AT PREVIOUS
C                   TIME STEP
C          X(I,2) = SAME AS ABOVE, EXCEPT AT CURRENT TIME STEP AND
C                   PREVIOUS NEWTON ITERATION
C          X(I,3) = SAME AS ABOVE, EXCEPT AT CURRENT NEWTON ITERATION
C         VX(I,1) = X-COMPONENT OF VELOCITY OF PARTICLE I
C         VX(I,2) =   WITH DEFINITION OF SECOND SUBSCRIPT
C         VX(I,3) =     SIMILAR TO THAT GIVEN FOR X ABOVE
C          Y(I,1) = SAME AS ABOVE
C          Y(I,2) =   EXCEPT FOR
C          Y(I,3) =     Y-COMPONENTS
C         VY(I,1) =       OF POSITION
C         VY(I,2) =         AND VELOCITY
C         VY(I,3) =           OF PARTICLE I

        OME1=1.-OMEGA
        DT2=.5*DT
        DO 8 I=1,3
        GM(I)=G*MASS(I)
        X(I,3)=X0(I)
        Y(I,3)=Y0(I)
        VX(I,3)=VX0(I)
      8 VY(I,3)=VY0(I)

C       COMPUTE INITIAL DISTANCES BETWEEN PARTICLES

        CALL RR(R(1,2))

C       PRINT OUT INITIAL PARTICLE POSITIONS

        CALL PRINT(0,X,Y)

C       BEGIN TIME STEP LOOP

        N=0
     10 N=N+1

C       UPDATE DISTANCES BETWEEN PARTICLES,PARTICLE POSITIONS, AND PAR-
C       TICLE VELOCITIES FOR PREVIOUS TIME STEP

        DO 12 I=1,3
        R(I,1)=R(I,2)
        X(I,1)=X(I,3)
        Y(I,1)=Y(I,3)
        VX(I,1)=VX(I,3)
     12 VY(I,1)=VY(I,3)

C       BEGIN NEWTON ITERATION LOOP

        DO 25 J=1,MAXIT

C       UPDATE PREVIOUS ITERATES FOR POSITIONS AND VELOCITIES AND COMPUTE
C       CURRENT ITERATES FOR POSITIONS

        DO 14 I=1,3
```

```
      X(I,2)=X(I,3)
      VX(I,2)=VX(I,3)
      X(I,3)=OME1*X(I,2)+OMEGA*(DT2*(VX(I,2)+VX(I,1))+X(I,1))
      Y(I,2)=Y(I,3)
      VY(I,2)=VY(I,3)
   14 Y(I,3)=OME1*Y(I,2)+OMEGA*(DT2*(VY(I,2)+VY(I,1))+Y(I,1))

C     COMPUTE DISTANCES AND FORCES BETWEEN PARTICLES FOR CURRENT
C     VALUES OF POSITION ITERATES

      CALL RR(R(1,2))
      CALL FXY

C     COMPUTE CURRENT ITERATES FOR VELOCITIES

      DO 16 I=1,3
      VX(I,3)=OME1*VX(I,2)+OMEGA*(DT  *FX(I)+VX(I,1))
   16 VY(I,3)=OME1*VY(I,2)+OMEGA*(DT  *FY(I)+VY(I,1))

C     TEST FOR CONVERGENCE OF NEWTON ITERATION

      DO 18 I=1,3
      IF(ABS(X(I,3)-X(I,2)).GT.EPS) GO TO 25
      IF(ABS(Y(I,3)-Y(I,2)).GT.EPS) GO TO 25
      IF(ABS(VX(I,3)-VX(I,2)).GT.EPS) GO TO 25
   18 IF(ABS(VY(I,3)-VY(I,2)).GT.EPS) GO TO 25
      GO TO 30
   25 CONTINUE

C     END OF NEWTON ITERATION LOOP

C     WRITE CONVERGENCE FAILURE MESSAGE AND GO TO NEXT DATA CASE

      WRITE(6,96) MAXIT
      GO TO 35

C     TEST FOR PRINTING OF POSITIONS AT CURRENT TIME STEP

   30 IF(MOD(N,INCPR).EQ.0) CALL PRINT(N,X,Y)

C     TEST FOR END OF TIME STEP LOOP FOR CURRENT DATA CASE

      IF(N.LT.NMAX) GO TO 10

C     END OF TIME STEP LOOP. TEST FOR LAST COMPUTATIONAL DATA CASE
C     FOR CURRENT PHYSICAL DATA.

   35 IF(END.GT.0.) GO TO 3
      GO TO 5

C     TERMINATION POINT FOR PROGRAM. CONTROL REACHES HERE UPON
C     ATTEMPTING TO READ PAST LAST DATA CARD.

   40 STOP
```

```
C     INTERNAL SUBROUTINE FOR COMPUTING DISTANCES BETWEEN PARTICLES

      SUBROUTINE RR(R)
      DIMENSION R(3)
      R(1)=SQRT((X(1,3)-X(2,3))**2+(Y(1,3)-Y(2,3))**2)
      R(2)=SQRT((X(1,3)-X(3,3))**2+(Y(1,3)-Y(3,3))**2)
      R(3)=SQRT((X(2,3)-X(3,3))**2+(Y(2,3)-Y(3,3))**2)
      RETURN

C     INTERNAL SUBROUTINE FOR COMPUTING FX(I),FY(I),I=1,2,3
C       WHERE  FX(I) = X-COMPONENT OF TOTAL FORCES ACTING ON PARTICLE I
C                      DIVIDED BY MASS OF PARTICLE I
C              FY(I) = SAME AS ABOVE WITH Y-COMPONENT

      SUBROUTINE FXY
      DIMENSION D(3)
      DO 2 I=1,3
    2 D(I)=R(I,1)*R(I,2)*(R(I,1)+R(I,2))
      TX1=(X(1,3)+X(1,1)-X(2,3)-X(2,1))/D(1)
      TY1=(Y(1,3)+Y(1,1)-Y(2,3)-Y(2,1))/D(1)
      TX2=(X(1,3)+X(1,1)-X(3,3)-X(3,1))/D(2)
      TY2=(Y(1,3)+Y(1,1)-Y(3,3)-Y(3,1))/D(2)
      TX3=(X(2,3)+X(2,1)-X(3,3)-X(3,1))/D(3)
      TY3=(Y(2,3)+Y(2,1)-Y(3,3)-Y(3,1))/D(3)
      FX(1)=-GM(2)*TX1-GM(3)*TX2
      FY(1)=-GM(2)*TY1-GM(3)*TY2
      FX(2)= GM(1)*TX1-GM(3)*TX3
      FY(2)= GM(1)*TY1-GM(3)*TY3
      FX(3)= GM(1)*TX2+GM(2)*TX3
      FY(3)= GM(1)*TY2+GM(2)*TY3
      RETURN

C     INTERNAL SUBROUTINE FOR PRINTING PARTICLE POSITIONS AT A SPECIFIED
C     TIME STEP

      SUBROUTINE PRINT(N,X,Y)
      DOUBLE PRECISION X(3,3),Y(3,3)
      REAL XR(3),YR(3)
      DO 5 I=1,3
      XR(I)=X(I,3)
    5 YR(I)=Y(I,3)
      WRITE(6,99) N,(XR(I),YR(I),I=1,3)
   99 FORMAT(1X I6,3X 3(2E13.6,3X))
      RETURN
      END
```